Chapter 1

Downward continuation

1.1 MIGRATION BY DOWNWARD CONTINUATION

Given waves observed along the earth’s surface, some well-known mathematical techniques that are introduced here enable us to extrapolate (downward continue) these waves down into the earth. Migration is a simple consequence of this extrapolation.

1.1.1 Huygens secondary point source

Waves on the ocean have wavelengths comparable to those of waves in seismic prospecting (15-500 meters), but ocean waves move slowly enough to be seen. Imagine a long harbor barrier parallel to the beach with a small entrance in the barrier for the passage of ships. This is shown in Figure 1.1. A plane wave incident on the barrier from the open ocean will send a wave through the gap in the barrier. It is an observed fact that the wavefront in the harbor becomes a circle with the gap as its center. The difference between this beam of water waves and a light beam through a window is in the ratio of wavelength to hole size.

Linearity is a property of all low-amplitude waves (not those foamy, breaking waves near the shore). This means that two gaps in the harbor barrier make two semicircular wavefronts. Where the circles cross, the wave heights combine by simple linear addition. It is interesting to think of a barrier with many holes. In the limiting case of very many holes, the barrier disappears, being nothing but one gap alongside another. Semicircular wavefronts combine to make only the incident plane wave. Hyperbolas do the same. Figure 1.2 shows hyperbolas increasing in density from left to right. All those waves at nonvertical angles must somehow combine with one another to extinguish all evidence of anything but the plane wave.

A Cartesian coordinate system has been superimposed on the ocean surface with x going along the beach and z measuring the distance from shore. For the analogy with reflection seismology, people are confined to the beach (the earth’s surface) where they make measurements of wave height as a function of x and t. From this data they can make inferences about the existence of gaps in the barrier out in the (x, z)-plane. The first frame of Figure 1.3 shows the 1

Figure 1.1: Waves going through a gap in a barrier have semicircular wavefronts (if the wavelength is long compared to the gap size). dwnc-storm [NR]

Figure 1.2: A barrier with many holes (top). Waves, (x,t)-space, seen beyond the barrier (bottom). dwnc-stormhole [ER]

1.1. MIGRATION BY DOWNWARD CONTINUATION 3

arrival time at the beach of a wave from the ocean through a gap. The earliest arrival occurs

Figure 1.3: The left frame shows the hyperbolic wave arrival time seen at the beach. Frames to the right show arrivals at increasing distances out in the water. The x-axis is compressed from Figure 1.1. dwnc-dc [ER]

nearest the gap. What mathematical expression determines the shape of the arrival curve seen in the (x,t)-plane?

The waves are expanding circles. An equation for a circle expanding with velocity v about a point (x_3, z_3) is

(x − x_3)^2 + (z − z_3)^2 = v^2 t^2 (1.1)

Considering t to be a constant, i.e. taking a snapshot, equation (1.1) is that of a circle. Considering z to be a constant, it is an equation in the (x,t)-plane for a hyperbola. Considered in the (t, x, z)-volume, equation (1.1) is that of a cone. Slices at various values of t show circles of various sizes. Slices of various values of z show various hyperbolas. Figure 1.3 shows four hyperbolas. The first is the observation made at the beach z_0 = 0. The second is a hypothetical set of observations at some distance z_1 out in the water. The third set of observations is at z_2, an even greater distance from the beach. The fourth set of observations is at z_3, nearly all the way out to the barrier, where the hyperbola has degenerated to a point. All these hyperbolas are from a family of hyperbolas, each with the same asymptote. The asymptote refers to a wave that turns nearly 90^∘ at the gap and is found moving nearly parallel to the shore at the speed dx/dt of a water wave. (For this water wave analogy it is presumed—incorrectly—that the speed of water waves is a constant independent of water depth).

If the original incident wave was a positive pulse, the Huygens secondary source must consist of both positive and negative polarities to enable the destructive interference of all but the plane wave. So the Huygens waveform has a phase shift. In the next section, mathematical expressions will be found for the Huygens secondary source. Another phenomenon, well known to boaters, is that the largest amplitude of the Huygens semicircle is in the direction pointing straight toward shore. The amplitude drops to zero for waves moving parallel to the shore. In optics this amplitude drop-off with angle is called the obliquity factor.

1.1.2 Migration derived from downward continuation

A dictionary gives many definitions for the word run. They are related, but they are distinct. Similarly, the word migration in geophysical prospecting has about four related but distinct meanings. The simplest is like the meaning of the word move. When an object at some location in the (x, z)-plane is found at a different location at a later time t, then we say it moves. Analogously, when a wave arrival (often called an event ) at some location in the (x,t)- space of geophysical observations is found at a different position for a different survey line at a greater depth z, then we say it migrates.

To see this more clearly, imagine the four frames of Figure 1.3 being taken from a movie. During the movie, the depth z changes beginning at the beach (the earth’s surface) and going out to the storm barrier. The frames are superimposed in Figure 1.4(left). Mainly what happens

Figure 1.4: Left shows a superposition of the hyperbolas of Figure 1.3. At the right the superposition incorporates a shift, called retardation t^′ = t + z/v, to keep the hyperbola tops together. dwnc-dcretard [ER]

in the movie is that the event migrates upward toward t = 0. To remove this dominating effect of vertical translation we make another superposition, keeping the hyperbola tops all in the same place. Mathematically, the time t axis is replaced by a so-called retarded time axis t^′ = t + z/v, shown in Figure 1.4(right). The second, more precise definition of migration is the motion of an event in (x,t^′)-space as z changes. After removing the vertical shift, the residual motion is mainly a shape change. By this definition, hyperbola tops, or horizontal layers, do not migrate.

The hyperbolas in Figure 1.4 really extend to infinity, but the drawing cuts each one off at a time equal √ 2 times its earliest arrival. Thus the hyperbolas shown depict only rays moving within 45^∘ of the vertical. It is good to remember this, that the ratio of first arrival time on a hyperbola to any other arrival time gives the cosine of the angle of propagation. The cutoff on each hyperbola is a ray at 45^∘. Notice that the end points of the hyperbolas on the drawing can be connected by a straight line. Also, the slope at the end of each hyperbola is the same. In physical space, the angle of any ray is tan θ = dx/dz. For any plane wave (or seismic event that is near a plane wave), the slope v dt/dx is sin θ, as you can see by considering a wavefront intercepting the earth’s surface at angle θ. So, energy moving on a straight line in physical (x, z)-space migrates along a straight line in data (x,t)-space. As z increases, the energy of all angles comes together to a focus. The focus is the exploding reflector. It is the gap in the barrier. This third definition of migration is that it is the process that somehow pushes observational data—wave height as a function of x and t —from the beach to the barrier.

The third definition stresses not so much the motion itself, but the transformation from the beginning point to the ending point.

To go further, a more general example is needed than the storm barrier example. The barrier example is confined to making Huygens sources only at some particular z. Sources are needed at other depths as well. Then, given a wave-extrapolation process to move data to increasing z values, exploding-reflector images are constructed with

Image (x, z) = Wave (t = 0, x, z) (1.2)

The fourth definition of migration also incorporates the definition of diffraction as the opposite of migration.

observations model

z = 0

all t migration −→ ←− diffraction t = 0

all z

Diffraction is sometimes regarded as the natural process that creates and enlarges hyperboloids. Migration is the computer process that does the reverse.

Another aspect of the use of the word migration arises where the horizontal coordinate can be either shot-to-geophone midpoint y, or offset h. Hyperboloids can be downward continued in both the (y,t)- and the (h,t)-plane. In the (y,t)-plane this is called migration or imaging, and in the (h,t)-plane it is called focusing or velocity analysis.

1.2 DOWNWARD CONTINUATION

Given a vertically upcoming plane wave at the earth’s surface, say u(t, x, z = 0) = u(t)const(x), and an assumption that the earth’s velocity is vertically stratified, i.e. v = v(z), we can presume that the upcoming wave down in the earth is simply time-shifted from what we see on the surface. (This assumes no multiple reflections.) Time shifting can be represented as a linear operator in the time domain by representing it as convolution with an impulse function. In the frequency domain, time shifting is simply multiplying by a complex exponential. This is expressed as

u(t, z) = u(t, z = 0) ∗ δ(t + z/v) (1.3) U(ω, z) = U(ω, z = 0) e^{−iωz/v} (1.4)

Sign conventions must be attended to, and that is explained more fully in chapter ??.

1.2.1 Continuation of a dipping plane wave.

Next consider a plane wave dipping at some angle θ. It is natural to imagine continuing such a wave back along a ray. Instead, we will continue the wave straight down. This requires the assumption that the plane wave is a perfect one, namely that the same waveform is observed at all x. Imagine two sensors in a vertical well bore. They should record the same signal except for a time shift that depends on the angle of the wave. Notice that the arrival time difference between sensors at two different depths is greatest for vertically propagating waves, and the time difference drops to zero for horizontally propagating waves. So the time shift �t is v^{−1} cosθ �z where θ is the angle between the wavefront and the earth’s surface (or the angle between the well bore and the ray). Thus an equation to downward continue the wave is

U(ω,θ, z + �z) = U(ω,θ, z) exp(−iω �t) (1.5)

U(ω,θ, z + �z) = U(ω,θ, z) exp ( −iω �z cos θ v ) (1.6)

Equation (1.6) is a downward continuation formula for any angle θ. Following methods of chapter ?? we can generalize the method to media where the velocity is a function of depth. Evidently we can apply equation (1.6) for each layer of thickness �z, and allow the velocity vary with z. This is a well known approximation that handles the timing correctly but keeps the amplitude constant (since |e^{iϕ}| = 1) when in real life, the amplitude should vary because of reflection and transmission coefficients. Suffice it to say that in practical earth imaging, this approximation is almost universally satisfactory.

In a stratified earth, it is customary to eliminate the angle θ which is depth variable, and change it to the Snell’s parameter p which is constant for all depths. Thus the downward continuation equation for any Snell’s parameter is

U(ω, p, z + �z) = U(ω, p, z) exp ( − iω�z v(z) √ 1 − p^2v(z)^2 ) (1.7)

It is natural to wonder where in real life we would encounter a Snell wave that we could downward continue with equation (1.7). The answer is that any wave from real life can be regarded as a sum of waves propagating in all angles. Thus a field data set should first be decomposed into Snell waves of all values of p, and then equation (1.7) can be used to downward continue each p, and finally the components for each p could be added. This process akin to Fourier analysis. We now turn to Fourier analysis as a method of downward continuation which is the same idea but the task of decomposing data into Snell waves becomes the task of decomposing data into sinusoids along the x-axis.

1.2.2 Downward continuation with Fourier transform

One of the main ideas in Fourier analysis is that an impulse function (a delta function) can be constructed by the superposition of sinusoids (or complex exponentials). In the study of time series this construction is used for the impulse response of a filter. In the study of functions of space, it is used to make a physical point source that can manufacture the downgoing waves that initialize the reflection seismic experiment. Likewise observed upcoming waves can be Fourier transformed over t and x.

Recall in chapter ??, a plane wave carrying an arbitrary waveform, specified by equation (??). Specializing the arbitrary function to be the real part of the function exp[−iω(t −t_0)] gives moving cosine wave = cos [ ω ( x v sin θ + z v cos θ − t ) ] (1.8)

Using Fourier integrals on time functions we encounter the Fourier kernel exp(−iωt). To use Fourier integrals on the space-axis x the spatial angular frequency must be defined. Since we will ultimately encounter many space axes (three for shot, three for geophone, also the midpoint and offset), the convention will be to use a subscript on the letter k to denote the axis being Fourier transformed. So k_x is the angular spatial frequency on the x-axis and exp(ik_x x) is its Fourier kernel. For each axis and Fourier kernel there is the question of the sign before the i. The sign convention used here is the one used in most physics books, namely, the one that agrees with equation (1.8). Reasons for the choice are given in chapter ??. With this convention, a wave moves in the positive direction along the space axes. Thus the Fourier kernel for (x, z,t)-space will be taken to be

Fourier kernel = e^{i kx x} e^{i kz z} e^{−iωt} = exp[i(k_x x + k_z z − ωt)] (1.9)

Now for the whistles, bells, and trumpets. Equating (1.8) to the real part of (1.9), physical angles and velocity are related to Fourier components. The Fourier kernel has the form of a plane wave. These relations should be memorized!

Angles and Fourier Components

sinθ = v k_x ω cosθ = v k_z ω (1.10)

A point in (ω,k_x ,k_z)-space is a plane wave. The one-dimensional Fourier kernel extracts frequencies. The multi-dimensional Fourier kernel extracts (monochromatic) plane waves.

Equally important is what comes next. Insert the angle definitions into the familiar relation sin^2 θ + cos^2 θ = 1. This gives a most important relationship:

k^2_x + k^2_z = ω^2 v^2

(1.11)

The importance of (1.11) is that it enables us to make the distinction between an arbitrary function and a chaotic function that actually is a wavefield. Imagine any function u(t, x, z). Fourier transform it to U(ω,k_x ,k_z). Look in the (ω,k_x ,k_z)-volume for any nonvanishing values of U. You will have a wavefield if and only if all nonvanishing U have coordinates that satisfy

(1.11). Even better, in practice the (t, x)-dependence at z = 0 is usually known, but the zdependence is not. Then the z-dependence is found by assuming U is a wavefield, so the z-dependence is inferred from (1.11).

Equation (1.11) also achieves fame as the “dispersion relation of the scalar wave equation,” a topic developed more fully in IEI.

1.2.3 Linking Snell waves to Fourier transforms

To link Snell waves to Fourier transforms we merge equations (??) and (??) with equations (1.10)

k_x ω = ∂t_0 ∂x = sin θ v = p (1.12)

k_z ω = ∂t_0 ∂z = cos θ v = √ 1 − p^2v^2 v (1.13)

The basic downward continuation equation for upcoming waves in Fourier space follows from equation (1.7) by eliminating p by using equation (1.12). For analysis of real seismic data we introduce a minus sign because equation (1.13) refers to downgoing waves and observed data is made from up-coming waves.

U(ω,k_x , z + �z) = U(ω,k_x , z) exp

− iω�z v √ 1 − v^2k^2_x ω^2

(1.14)

In Fourier space we delay signals by multiplying by e^{iω�t} , analogously, equation (1.14) says we downward continue signals into the earth by multiplying by e^{ikz �z}. Multiplication in the Fourier domain means convolution in time which can be depicted by the engineering diagram in Figure 1.5.

Figure 1.5: Downward continuation of a downgoing wavefield. dwnc-inout [NR]

Downward continuation is a product relationship in both the ω-domain and the k_x -domain. Thus it is a convolution in both time and x. What does the filter look like in the time and space domain? It turns out like a cone, that is, it is roughly an impulse function of x^2 + z^2 − v^2t^2. More precisely, it is the Huygens secondary wave source that was exemplified by ocean waves entering a gap through a storm barrier. Adding up the response of multiple gaps in the barrier would be convolution over x.

A nuisance of using Fourier transforms in migration and modeling is that spaces become periodic. This is demonstrated in Figure 1.6. Anywhere an event exits the frame at a side, top, or bottom boundary, the event immediately emerges on the opposite side. In practice, the unwelcome effect of periodicity is generally ameliorated by padding zeros around the data and the model.

Figure 1.6: A reflectivity model on the left and synthetic data using a Fourier method on the right. dwnc-diag [ER]

1.3 PHASE-SHIFT MIGRATION

The phase-shift method of migration begins with a two-dimensional Fourier transform (2D- FT) of the dataset. (See chapter ??.) This transformed data is downward continued with exp(ik_z z) and subsequently evaluated at t = 0 (where the reflectors explode). Of all migration methods, the phase-shift method most easily incorporates depth variation in velocity. The phase angle and obliquity function are correctly included, automatically. Unlike Kirchhoff methods, with the phase-shift method there is no danger of aliasing the operator. (Aliasing the data, however, remains a danger.)

Equation (1.14) referred to upcoming waves. However in the reflection experiment, we also need to think about downgoing waves. With the exploding-reflector concept of a zerooffset section, the downgoing ray goes along the same path as the upgoing ray, so both suffer the same delay. The most straightforward way of converting one-way propagation to two-way propagation is to multiply time everywhere by two. Instead, it is customary to divide velocity everywhere by two. Thus the Fourier transformed data values, are downward continued to a depth �z by multiplying by

e^{i kz �z} = exp

− i 2ω v √ 1 − v^2 k^2_x 4ω^2 �z (1.15)

Ordinarily the time-sample interval �τ for the output-migrated section is chosen equal to the time-sample rate of the input data (often 4 milliseconds). Thus, choosing the depth �z = (v/2)�τ , the downward-extrapolation operator for a single time step �τ is

C = exp

−i ω �τ √ 1 − v^2 k^2_x 4ω^2

(1.16)

Data will be multiplied many times by C, thereby downward continuing it by many steps of �τ .

1.3.1 Pseudocode to working code

Next is the task of imaging. Conceptually, at each depth an inverse Fourier transform is followed by selection of its value at t = 0. (Reflectors explode at t = 0). Since only the Fourier transform at one point, t = 0, is needed, other times need not be be computed. We know the ω = 0 Fourier component is found by the sum over all time, analogously, the t = 0 component is found as the sum over all ω. (This is easily shown by substituting t = 0 into the inverse Fourier integral.) Finally, inverse Fourier transform k_x to x. The migration process, computing the image from the upcoming wave u, may be summarized in the following pseudo code:

U(ω,k_x ,τ = 0) = FT [u(t, x)] For τ = �τ ,2�τ , . . . , end of time axis on seismogram For all k_x

For all ω

C = exp(−iω�τ √ 1 − v^2k^2_x /4ω^2) U(ω,k_x ,τ ) = U(ω,k_x ,τ − �τ ) ∗ C For all k_x

Image(k_x ,τ ) = 0. For all ω

Image(k_x ,τ ) = Image(k_x ,τ ) +U(ω,k_x ,τ ) image(x,τ ) = FT [Image(k_x ,τ )]

This pseudo code Fourier transforms a wavefield observed at the earth’s surface τ = 0, and then it marches that wavefield down into the earth (τ > 0) filling up a three-dimensional function, U(ω,k_x ,τ ). Then it selects t = 0, the time of the exploding reflectors by summing over all frequencies ω. (Mathematically, this is like finding the signal at ω = 0 by summing over all t).

Turning from pseudocode to real code, an important practical reality is that computer memories are not big enough for the three-dimensional function U(ω,k_x ,τ ). But it is easy to intertwine the downward continuation with the summation over ω so a three-dimensional

1.3. PHASE-SHIFT MIGRATION 11

function need not be kept in memory. This is done in the real code in subroutine phasemig().

subroutine phasemig( up, nt, nx, dt, dx, image, ntau, dtau, vel)
integer nt, nx, ntau, iw,nw,ikx,itau
real dt,dx, w,w0,dw, kx,kx0,dkx,dtau, vel, sig1,sig2,pi, w2, vkx2
complex up(nt,nx), image(ntau,nx), cc
                        pi = 3.14159265; sig1 = +1.; sig2 = -1.
call ft1axis( 0, sig1, nt, nx, up)
call ft2axis( 0, sig2, nt, nx, up)
        nw = nt; w0 = -pi/dt; dw = 2.*pi/(nt*dt)
                             kx0 = -pi/dx; dkx= 2.*pi/(nx*dx)
call null( image, ntau*nx*2)
do iw = 2, nw { w = w0 + (iw -1) * dw
do ikx = 2, nx { kx = kx0 + (ikx-1) * dkx
        w2 = w * w
        vkx2 = vel*vel * kx*kx / 4.
        if( w2 > vkx2 ) {
                cc = cexp( cmplx( 0., - w * dtau * sqrt( 1. - vkx2/w2 ) ) )
                do itau = 1, ntau {
                        up(iw,ikx) = up(iw,ikx) * cc
                        image(itau,ikx) = image(itau,ikx) + up(iw,ikx)
                        }
                }
        }}
call ft2axis( 1, sig2, ntau, nx, image)
return; end

Conjugate migration (modeling) proceeds in much the same way. Beginning from an upcoming wave that is zero at great depth, the wave is marched upward in steps by multiplication with exp(ik_z�z). As each level in the earth is passed, exploding reflectors from that level are added into the upcoming wave. Pseudo code for modeling the upcoming wave u is

Image(k_x , z) = FT [image(x, z)] For all ω and all k_x

U(ω,k_x ) = 0. For all ω { For all k_x { For z = z_{max}, z_{max} − �z, z_{max} − 2�z, . . . , 0 { C = exp(+i�zω √ v^{−2} − k_x^2/ω^2) U(ω,k_x ) = U(ω,k_x ) ∗ C U(ω,k_x ) = U(ω,k_x )+Image(k_x , z) } } } u(t, x) = FT [U(ω,k_x )]

Some real code for this job is in subroutine phasemod().

subroutine phasemod( image, nz, nx, dz, dx, up, nt, dt, vel)
integer nz, nx, nt, iw,nw,ikx,iz
real dt,dx,dz, w,w0,dw, kx,kx0,dkx, vel, sig1,sig2,pi, w2, vkx2
complex up(nt,nx), image(nz,nx), cc
                pi = 3.14159265; sig1 = +1.; sig2 = -1.
call ft2axis( 0, sig2, nz, nx, image)
        nw = nt; w0 = -pi/dt; dw = 2.*pi/(nt*dt)
                        kx0 = -pi/dx; dkx= 2.*pi/(nx*dx)
call null( up, nw*nx*2)
do iw = 2, nw { w = w0 + (iw-1) * dw
do ikx = 2, nx { kx = kx0 + (ikx-1) * dkx
        w2 = w * w
        vkx2 = vel*vel * kx*kx / 4.
        if( w2 > vkx2 ) {
                cc = cexp( cmplx( 0., w * dz * sqrt(1. - vkx2/w2) ))
                do iz = nz, 1, -1
                        up(iw,ikx) = up(iw,ikx) * cc + image(iz,ikx)
                }
        }}
call ft1axis( 1, sig1, nt, nx, up)
call ft2axis( 1, sig2, nt, nx, up)
return; end

The positive sign in the complex exponential is a combination of two negatives, the up coming wave and the upward extrapolation. In principle, the three loops on ω, k_x , and z are interchangeable, however, since this tutorial program uses a velocity v that is a constant function of depth, I speeded it by a large factor by putting the z-loop on the inside and pulling the complex exponential out of the inner loop. Figure 1.2 was made with subroutine phasemod() on the preceding page.

1.3.2 Kirchhoff versus phase-shift migration

In chapter ??, we were introduced to the Kirchhoff migration and modeling method by means of subroutines kirchslow() on page ?? and kirchfast() on page ??. From chapter ?? we know that these routines should be supplemented by a √ −iω filter such as subroutine

halfdifa() on page ??. Here, however, we will compare results of the unadorned subroutine kirchfast() on page ?? with our new programs, phasemig() on the page before and phasemod() on the preceding page. Figure 1.7 shows the result of modeling data and then migrating it. Kirchhoff and phase-shift migration methods both work well. As expected, the Kirchhoff method lacks some of the higher frequencies that could be restored by √ −iω. Another problem is the irregularity of the shallow bedding. This is an operator aliasing problem addressed in chapter 10. Figure 1.8 shows the temporal spectrum of the original sigmoid model, along with the spectrum of the reconstruction via phase-shift methods. We see the spectra are essentially identical with little growth of high frequencies as we noticed with the Kirchhoff method in Figure ??. Figure 1.9 shows the effect of coarsening the space axis. Synthetic data is generated from an increasingly subsampled model. Again we notice that the phase-shift method of this chapter produces more plausible results than the simple Kirchhoff programs of chapter ??.

Figure 1.7: Reconstruction after modeling. Left is by the nearest-neighbor Kirchhoff method. Right is by the phase shift method. dwnc-comrecon [ER,M]

Figure 1.8: Top is the temporal spectrum of the model. Bottom is the spectrum of the reconstructed model. dwnc-phaspec [ER]

1.3.3 Damped square root

The definition of k_z as k_z = √ ω^2/v^2 − k^2_x obscures two aspects of k_z. First, which of the two square roots is intended, and second, what happens when k^2_x > ω^2/v^2. For both coding and theoretical work we need a definition of ik_z that is valid for both positive and negative values of ω and for all k_x . Define a function R = ik_z(ω,k_z) by

R = ik_z = √ (−iω + ϵ)^2 + k^2_x (1.17)

It is important to know that for any ϵ > 0, and any real ω and real k_x that the real part ℜR > 0 is positive. This means we can extrapolate waves safely with e^{−Rz} for increasing z or with e^{+Rz} for decreasing z. To switch from downgoing to upcoming we use the complex conjugate R. Thus we have disentangled the damping from the direction of propagation.

Let us see why ℜR > 0 is positive for all real values of ω and k_x . Recall that for ω ranging between ±∞, e^{iω�t} rotates around the unit circle in the complex plane. Examine Figure 1.10 which shows the complex functions:

1. f (ω) = ϵ − iω,

2. −iω̂ = (1 + ϵ) − e^{iω�t} ,

Figure 1.9: Modeling with increasing amounts of lateral subsampling. Left is the nearestneighbor Kirchhoff method. Right is the phase-shift method. Top has 200 channels, middle has 100 channels, and bottom has 50 channels. dwnc-commod [ER]

3. (−iω̂)^2,

4. (ik_z)^2 = (−iω̂)^2 + k^2_x , and

5. ik_z = [(−iω̂)^2 + k^2_x ]^{1/2}

Figure 1.10: Some functions in the complex plane. dwnc-francis [ER]

The first two panels are explained by the first two functions. The first two functions and the first two panels look different but they become the same in the practical limit of ϵ → 0 and �t → 0. The left panel represents a time derivative in continuous time, and the second panel likewise in sampled time is for a “causal finite-difference operator” representing a time derivative. Notice that the graphs look the same near ω = 0. As we sample seismic data with increasing density, �t → 0, the frequency content shifts further away from the Nyquist frequency. Measuring ω in radians/sample, in the limit �t → 0, the physical energy is all near ω = 0.

The third panel in Figure 1.10 shows (−iω̂)^2 which is a cardioid that wraps itself close up to the negative imaginary axis without touching it. (To understand the shape near the origin, think about the square of the leftmost plane. You may have seen examples of the negative imaginary axis being a branch cut.) In the fourth panel a small positive quantity k^2_x is added which shifts the cardioid to the right a bit. Taking the square root gives the last panel which shows the curve in the right half plane thus proving the important result we need, that ℜ ik_z(ω,k_z) > 0 for all real ω. It is also positive for all real k_x because any k^2_x > 0 shifts the cardioid to the right. The additional issue of time causality in forward modeling is covered in IEI.

Luckily the Fortran csqrt() function assumes the phase of the argument is between ±180^∘ exactly as we need here. Thus the square root itself will have a phase between ±90^∘ as we require. In applications, ϵ would typically be chosen proportional to the maximum time on the data. Thus the mathematical expression −iω + ϵ might be rendered in Fortran as cmplx(qi,-omega) where qi=1./tmax and the whole concept implemented as in function eiktau() on the current page. Do not set qi=0 because then the csqrt() function cannot decipher positive from negative frequencies.

complex function eiktau( dt, w, vkx, qi )
real dt, w, vkx, qi
eiktau = cexp( - dt * csqrt( cmplx( qi, -w) ** 2 + vkx * vkx /4. ) )
return; end

Finally, you might ask, why bother with all this careful theory connected with the damped square root. Why not simply abandon the evanescent waves as done by the “if” statement in subroutines phasemig() and phasemod()? There are several reasons:

1. The exploding reflector concept fails for evanescent waves (when ω^2 < v^2k^2_x ). Realistic modeling would have them damping with depth. Rather than trying to handle them correctly we will make a choice, either (1) to abandon evanescent waves effectively setting them to zero, or (2) we will take them to be damping. (You might notice that when we switch from downgoing to upgoing, a damping exponential switches to a growing exponential, but when we consider the adjoint of applying a damped exponential, that adjoint is also a damped exponential.)

I’m not sure if there is a practical difference between choosing to damp evanescent waves or simply to set them to zero, but there should be a noticable difference on synthetic data: When a Fourier-domain amplitude drops abruptly from unity to zero, we can expect a time-domain signal that spreads widely on the time axis, perhaps dropping off slowly as 1/t. We can expect a more concentrated pulse if we include the evanescent energy, even though it is small. I predict the following behavior: Take an impulse; diffract it and then migrate it. When evanescent waves have been truncated, I predict the impulse is turned into a “butterfly” whose wings are at the hyperbola asymptote. Damping the evanescent waves, I predict, gives us more of a “rounded” impulse.

2. In a later chapter we will handle the x-axis by finite differencing (so that we can handle v(x). There a stability problem will develop unless we begin from careful definitions as we are doing here.

3. Seismic theory includes an abstract mathematical concept known as branch-line integrals. Such theory is most easily understood beginning from here.

1.3.4 Adjointness and ordinary differential equations

It is straightforward to adapt the simple programs phasemig() on page 11 and phasemod() on page 11 to depth variable velocity. As you might suspect, the two processes are adjoint to each other, and for reasons given at the end of chapter ?? it is desirable to code them to be so. With differential equations and their boundary conditions, the concept of adjoint is more subtle than previous examples. Thus, I postponed till here the development of adjoint code for phase-shift migration. This coding is a little strenuous, but it affords a review of many basic concepts, so we do so here. (Feel free to skip this section.) It is nice to have a high quality code for this fundamental operation.

Many situations in physics are expressed by the differential equation du dz − iα u = s(z) (1.18)

In the migration application, u(z) is the up-coming wave, α = − √ ω^2/v^2 − k^2_x , s(z) is the exploding-reflector source. We take the medium to be layered (v constant in layers) so that α is constant in a layer, and we put the sources at the layer boundaries. Thus within a layer we have du/dz − iα u = 0 which has the solution

u(z_k + �z) = u(z_k ) e^{iα�z} (1.19)

For convenience, we use the “delay operator” in the k-th layer Z_k = e^{−iα�z} so the delay of upward propagation is expressed by u(z_k) = Z_k u(z_k + �z). (Since α is negative for upcoming waves, Z_k = e^{−iα�z} has a positive exponent which represents delay.) Besides crossing layers, we must cross layer boundaries where the (reflection) sources add to the upcoming wave. Thus we have u_{k−1} = Z_{k−1}u_k + s_{k−1} (1.20)

Recursive use of equation (1.20) across a medium of three layers is expressed in matrix form as

Mu =

1 −Z_0 . . . 1 −Z_1 . . . 1 −Z_2 . . . 1 u_0 u_1 u_2 u_3 = s_0 s_1 s_2 s_3

= s (1.21)

A recursive solution begins at the bottom with u_3 = s_3 and propagates upward.

The adjoint (complex conjugate) of the delay operator Z is the time advance operator Z̄. The adjoint of equation (1.21) is given by

M^′ s̃ =

1 . . . −Z̄_0 1 . .

. −Z̄_1 1 . . . −Z̄_2 1 s̃_0 s̃_1 s̃_2 s̃_3 = u_0 u_1 u_2 u_3

= u (1.22)

where s̃(z) (summed over frequency) is the migrated image. The adjointness of equation (1.21) and (1.22) seems obvious, but it is not the elementary form we are familiar with because the matrix multiplies the output (instead of multiplying the usual input). To prove the adjointness, notice that equation (1.21) is equivalent to u = M^{−1}s whose adjoint, by definition, is s̃ = (M^{−1})^′u which is s̃ = (M^′)^{−1}u (because of the basic mathematical fact that the adjoint of an inverse is the inverse of the adjoint) which gives M^′s̃ = u which is equation (1.22).

We observe the wavefield only on the surface z = 0, so the adjointness of equations (1.21) and (1.22) is academic because it relates the wavefield at all depths with the source at all depths. We need to truncate u to its first coefficient u_0 since the upcoming wave is known only at the surface. This truncation changes the adjoint in a curious way. We rewrite equation (1.21) using a truncation operator T that is the row matrix T = [1,0,0,···] getting u_0 = Tu = TM^{−1}s. Its adjoint is ŝ = (M^{−1})^′T^′u^′_0 = (M^′)^{−1}T^′u^′_0 or M^′ŝ = T^′u_0 which looks like

M^′ s̃ =

1 . . . −Z̄_0 1 . .

. −Z̄_1 1 . . . −Z̄_2 1 s̃_0 s̃_1 s̃_2 s̃_3 = u_0 0 0 0 (1.23)

18 CHAPTER 1. DOWNWARD CONTINUATION

The operator 1.23 is a recursion beginning from s̃_0 = u_0 and continuing downward with

s̃_k = Z̄_{k−1} s̃_{k−1} (1.24)

A final feature of the migration application is that the image is formed from s̃ by summing over all frequencies. Although I believe the mathematics above and the code in subroutine

gazadj() on this page, I ran the dot product test to be sure!

# Phase shift modeling and migration. (Warning: destroys its input!)
#
subroutine gazadj( adj, dt,dx, v,nt,nx, modl, data )
integer adj, nt,nx, iw, ikx, iz,nz
complex eiktau, cup, modl(nt,nx), data(nt,nx)
real dt,dx, v(nt), pi, w,w0,dw, kx,kx0,dkx,qi
call adjnull( adj, 0, modl,nt*nx*2, data,nt*nx*2 )
pi = 4.*atan(1.); w0 = -pi/dt; dw = 2.*pi/(nt*dt); qi=.5/(nt*dt)
nz = nt; kx0 = -pi/dx; dkx= 2.*pi/(nx*dx)
if( adj == 0) call ft2axis( 0, -1., nz, nx, modl)
else { call ft2axis( 0, -1., nt, nx, data)
                call ft1axis( 0, 1., nt, nx, data)
                }
do ikx = 2, nx { kx = kx0 + (ikx-1) * dkx
do iw = 2, 1+nt/2 { w = w0 + (iw -1) * dw
        if( adj== 0) { data(iw,ikx) = modl(nz,ikx)
                do iz = nz-1, 1, -1
                        data(iw,ikx) = data(iw,ikx) * eiktau(dt,w,v(iz)*kx,qi) +
                                       modl(iz,ikx)
                }
        else { cup = data(iw,ikx)
                do iz = 1, nz {
                        modl(iz,ikx) = modl(iz,ikx) + cup
                        cup = cup * conjg( eiktau(dt,w,v(iz)*kx,qi))
                        }
                }
        }}
if( adj == 0) { call ft1axis( 1, 1., nt, nx, data)
                  call ft2axis( 1, -1., nt, nx, data) }
else { call ft2axis( 1, -1., nz, nx, modl) }
return; end

Finally, a few small details about the code. The loop on spatial frequency ikx begins at ikx=2. The reason for the 2, instead of a 1, is to omit the Nyquist frequency. If the Nyquist frequency were to be included, it should be divided into one half at positive Nyquist and one half at negative Nyquist, which would clutter the code without adding practical value. Another small detail is that the loop on temporal frequency iw begins at iw=1+nt/2 which effectly omits negative frequencies. This is purely an economy measure. Including the negative frequencies would assure that the final image be real, no imaginary part. Omitting negative frequencies simply gives an imaginary part that can be thrown away, and gives the same real image, scaled by a half. The factor of two speed up makes these tiny compromises well worthwhile.

1.3.5 Vertical exaggeration example

To examine questions of vertical exaggeration and spatial resolution we consider a line of point scatters along a 45^∘ dipping line in (x, z)-space. We impose a linear velocity gradient such as that typically found in the Gulf of Mexico, i.e. v(z) = v_0 + αz with α = 1/2s^{−1}. Viewing our point scatterers as a function of traveltime depth, τ = 2

∫_z 0 dz/v(z) in Figure 1.11 we see, as expected, that the points, although separated by equal intervals in x, are separated by shorter time intervals with increasing depth. The points are uniformly separated along a straight line in (x, z)-space, but they are nonuniformly separated along a curved line in (x,τ )- space. The curve is steeper near the earth’s surface where v(z) yields the greatest vertical exaggeration. Here the vertical exaggeration is about unity (no exageration) but deeper the vertical exaggeration is less than unity (horizontal exaggeration). Applying subroutine gazadj() on

Figure 1.11: Points along a 45 degree slope as seen as a function of traveltime depth. dwnc-sagmod [ER]

the preceding page the points spray out into hyperboloids (like hyperbolas, but not exactly) shown in Figure 1.12. The obvious feature of this synthetic data is that the hyperboloids appear to have different asymptotes. In fact, there are no asymptotes because an asymptote is

Figure 1.12: The points of Figure 1.11 diffracted into hyperboloids. dwnc-sagdat [ER]

a ray going horizontal at a more-or-less constant depth, which will not happen in this model because the velocity increases steadily with depth.

(I should get energetic and overlay these hyperboloids on top of the exact hyperbolas of the Kirchhoff method, to see if there are perceptible traveltime differences.)

1.3.6 Vertical and horizontal resolution

In principle, migration converts hyperbolas to points. In practice, a hyperbola does not collapse to a point, it collapses to a focus. A focus has measurable dimensions. Vertical resolution is easily understood. For a given frequency, higher velocity gives longer vertical wavelength and thus less resolution. When the result of migration is plotted as a function of traveltime depth τ instead of true depth z, however, enlargement of focus with depth is not visible.

Horizontal resolution works a little differently. Migration is said to be “good” because it increases spatial resolution. It squeezes a large hyperbola down to a tiny focus. Study the focal widths in Figure 1.13. Notice the water-velocity focuses hardly broaden with depth. We expect some broadening with depth because the late hyperbolas are cut off at their sides and bottom (an aperture effect), but there is no broadening here because the periodicity of the Fourier domain means that events are not truncated but wrapped around.

Figure 1.13: Left is migration back to a focus with a constant, water-velocity model. Right is the same, but with a Gulf of Mexico velocity, i.e. the hyperboloids of Figure 1.12 migrated back to focuses. Observe focus broadening with depth. dwnc-sagres [ER]

When the velocity increases with depth, wider focuses are found at increasing depth. Why is this? Consider each hyperbola to be made of many short plane wave segments. Migration moves all the little segments on top of each other. The sharpness of a focus cannot be narrower than the width of each of the many plane-wave segments that superpose to make the focus. The narrowest of these plane-wave segments is at the steepest part of a hyperbola asymptote. Deeper reflectors (which have later tops) have less steep asymptotes because of the increasing velocity. Thus deeper reflectors with faster RMS velocities have wider focuses so the deeper part of the image is more blurred. A second way to understand increased blurring with depth is from equation (1.12), that the horizontal frequency k_x = ωp = ωv^{−1} sinθ is independent of depth. The steepest possible angle occurs when |sinθ| = 1. Thus, considering all possible angles, the largest |k_x | is |k_x | = |ω|/v(z). Larger values of horizontal frequency |k_x | could help us get narrower focuses, but the deeper we go (faster velocity we encounter), the more these high frequencies are lost because of the evanescent limit |k_x | ≤ |ω/v(z)|. The limit is where the ray goes no deeper but bends around and comes back up again without ever reflecting. Any ray that does this many times is said to be a surface-trapped wave. It cannot sharpen a deep focus.

1.3.7 Field data migration

Application of subroutine gazadj() on page 18 to the Gulf of Mexico data set processed in earlier chapters yields the result in Figure 1.14.

EXERCISES:

1 Devise a mathematical expression for a plane wave that is an impulse function of time with a propagation angle of 15^∘ from the vertical z-axis in the plus z direction. Express the result in the domain of

(a) (t, x, z)

(b) (ω, x, z)

(c) (ω,k_x , z)

(d) (ω, p, z)

Figure 1.14: Phase shift migration of Figure ??. Press button for movie to compare to stack and Kirchhoff migration of Figure ??. dwnc-wgphase [ER,M]

Index

damped square root, 13 delay operator, 17 diffraction, 5 dip, 6 downward continue, 1

eiktau subroutine, 15

focus, 5 Fourier downward continuation, 6

gazadj subroutine, 17

migration

phase-shift, 9 migration, defined, 5

phasemig subroutine, 11 phasemod subroutine, 11

recursion, downward continuation, 17 resolution, 18, 20

Snell wave, 6, 8 subroutine

eiktau, exp ikz, 15
gazadj, phase shift mig., 17
phasemig, migration, 11
phasemod, diffraction, 11

vertical exaggeration, 18

wave equation, 8

215

216 INDEX